

# NAG Fortran Library Routine Document

## C05ADF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

C05ADF locates a zero of a continuous function in a given interval by a combination of the methods of linear interpolation, extrapolation and bisection.

### 2 Specification

```
SUBROUTINE C05ADF(A, B, EPS, ETA, F, X, IFAIL)
INTEGER          IFAIL
real           A, B, EPS, ETA, F, X
EXTERNAL        F
```

### 3 Description

The routine attempts to obtain an approximation to a simple zero of the function  $f(x)$  given an initial interval  $[a, b]$  such that  $f(a) \times f(b) \leq 0$ . The zero is found by calls to C05AZF whose specification should be consulted for details of the method used.

The approximation  $x$  to the zero  $\alpha$  is determined so that one or both of the following criteria are satisfied:

- (i)  $|x - \alpha| < \text{EPS}$ ,
- (ii)  $|f(x)| < \text{ETA}$ .

### 4 References

None.

### 5 Parameters

- 1:  $A$  – *real* *Input*  
*On entry:* the lower bound of the interval,  $a$ .
- 2:  $B$  – *real* *Input*  
*On entry:* the upper bound of the interval,  $b$ .  
*Constraint:*  $B \neq A$ .
- 3:  $\text{EPS}$  – *real* *Input*  
*On entry:* the absolute tolerance to which the zero is required (see Section 3).  
*Constraint:*  $\text{EPS} > 0.0$ .
- 4:  $\text{ETA}$  – *real* *Input*  
*On entry:* a value such that if  $|f(x)| < \text{ETA}$ ,  $x$  is accepted as the zero.  $\text{ETA}$  may be specified as 0.0 (see Section 7).
- 5:  $F$  – *real* FUNCTION, supplied by the user. *External Procedure*  
 $F$  must evaluate the function  $f$  whose zero is to be determined.

Its specification is:

<pre> <b>real</b> FUNCTION F(XX) <b>real</b>          XX  1:  XX – <i>real</i>                                     <i>Input</i>      <i>On entry:</i> the point at which the function must be evaluated. </pre>
---

F must be declared as EXTERNAL in the (sub)program from which C05ADF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

6: X – *real* *Output*

*On exit:* the approximation to the zero.

7: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $EPS \leq 0.0$ ,  
 or  $A = B$ ,  
 or  $F(A) \times F(B) > 0.0$ .

IFAIL = 2

Too much accuracy has been requested in the computation, that is, EPS is too small for the computer being used. The final value of X is an accurate approximation to the zero.

IFAIL = 3

A change in sign of  $f(x)$  has been determined as occurring near the point defined by the final value of X. However, there is some evidence that this sign-change corresponds to a pole of  $f(x)$ .

IFAIL = 4

Indicates that a serious error has occurred in C05AZF. Check all routine calls. Seek expert help.

## 7 Accuracy

This depends on the value of EPS and ETA. If full machine accuracy is required, they may be set very small, resulting in an error exit with IFAIL = 2, although this may involve many more iterations than a lesser accuracy. The user is recommended to set  $ETA = 0.0$  and to use EPS to control the accuracy, unless he has considerable knowledge of the size of  $f(x)$  for values of  $x$  near the zero.

## 8 Further Comments

The time taken by the routine depends primarily on the time spent evaluating F (see Section 5).

If it is important to determine an interval of length less than EPS containing the zero, or if the function F is expensive to evaluate and the number of calls to F is to be restricted, then use of C05AZF is recommended. Use of C05AZF is also recommended when the structure of the problem to be solved does not permit a simple function F to be written: the reverse communication facilities of C05AZF are more flexible than the direct communication of F required by C05ADF.

## 9 Example

The example program below calculates the zero of  $e^{-x} - x$  within the interval  $[0, 1]$  to approximately 5 decimal places.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      C05ADF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER       (NOUT=6)
*      .. Local Scalars ..
      real            A, B, EPS, ETA, X
      INTEGER          IFAIL
*      .. External Functions ..
      real            F
      EXTERNAL         F
*      .. External Subroutines ..
      EXTERNAL         C05ADF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'C05ADF Example Program Results'
      A = 0.0e0
      B = 1.0e0
      EPS = 1.0e-5
      ETA = 0.0e0
      IFAIL = 1
*
      CALL C05ADF(A,B,EPS,ETA,F,X,IFAIL)
*
      WRITE (NOUT,*)
      IF (IFAIL.EQ.0) THEN
        WRITE (NOUT,99999) 'Zero =', X
      ELSE
        WRITE (NOUT,99998) 'IFAIL =', IFAIL
        IF (IFAIL.EQ.2 .OR. IFAIL.EQ.3) WRITE (NOUT,99999)
+      'Final point = ', X
      END IF
      STOP
*
99999 FORMAT (1X,A,F12.5)
99998 FORMAT (1X,A,I3)
      END
*
      real FUNCTION F(X)
*      .. Scalar Arguments ..
      real            X
*      .. Intrinsic Functions ..
      INTRINSIC       EXP
*      .. Executable Statements ..
      F = EXP(-X) - X
      RETURN
      END
```

## 9.2 Program Data

None.

## 9.3 Program Results

C05ADF Example Program Results

Zero = 0.56714

---